
Approach

I have written this programme using semantic grammar. Semantic grammar is most appropriate when you are only dealing with a small set of possible phrases. This is exactly the sort of situation we are in with a hotel booking system. There aren't that many different questions that can be asked so they are quite easy to represents by using semantic grammar.

In defining the grammar the most important part was the room descriptions. Almost all sentences used in this programme include a reference to a type of room. Rooms can either be single or double and can be with a shower or without a shower. In this program rooms can also been requested for or inquired about for tonight and tomorrow night. Because of this when defining the room predicate there are three values returned, roomtype1 (single or double), roomtype2 (with a shower or without a shower returned as yes and no) and night (tonight and tomorrow night returned as one and two). So in a sentence "Please book me a single room without shower for tonight ." the values returned would be single, no, 1. A question is not always that specific though the customer may just say "Please book me a room ." in this case we do not know what type of room and when the customer would like it so the values returned are 0, 0, 0. In the case of booking a room this system will then have to ask the customer if they require a single or double room, if they require a shower and which night they would like to book the room for. In other situations such as "Do you have a room" it may not be necessary to ask for this information as the customer is not interested in what type of room they just want to know if we have any room so any rooms will do.

In defining grammar I have used several blank lists. For example roomtype1 has the following listing:

roomtype1(single) --> [single].

roomtype1(double) --> [double].

roomtype1(0) --> [].

This means that when roomtype1 is used to make up the room predicate I do not have to define two predicates, one with roomtype1 and one without roomtype1 as roomtype1 can be blank. I have used this in several other places as well such as the polite predicate, questions may be preceded by “Please” or “Please would you” but may not. Because of this I have included a blank list as one of the Please predicates so that I can put the Please predicate at the beginning of all appropriate sentences and not have to worry about duplicating these sentences, one copy with the Please predicate and one without.

When the sentence has been successfully recognise we have to do something with it and give the user an answer. If a sentence is not recognised the reply predicate fails and an alternative reply predicate succeeds which prints out "Please rephrase the question.". If the sentence is recognised it fires an appropriate predicate to deal with the input. In the example of booking a room the bookroom predicate fires and is sent the values of roomtype1, roomtype2, night and number of rooms. If any of these values are zero indicating that a value was not specified the predicate asks the user to specify exactly what type of room they require. When all required information has been gathered the predicate checks that a room is available. If a room is available it asks the customer for their name and proceeds to book the number of rooms that had been requested as long as rooms remain available. For each room booked it tells the customer which room they have been booked into and when.

When a room is booked for a customer their name is recorded in a dynamic predicate called customer. Another dynamic predicate called roombooking is retracted and another one asserted with the customer's name in the room booked field. If the room booked field reads roomfree this indicates that the room has not been booked. It is possible that a customer with the surname roomfree books a room which will cause problems but this is extremely unlikely. The roombooking predicate has three fields room number, day and customer. Currently the day field only holds one or two referring to today or tomorrow but this could easily be extended to hold a date instead if this programme were to be used in the real-world. The customer predicate also has three fields, one for title, one for name and one for booked in. The title field holds a number referring to the title predicate, when a name is entered it has to conform to title followed by name. If an invalid title has been entered the customer is it repeatedly asked to enter their name until a valid one is encountered.
The booked in field States whether the customer has booked in yet. This can be updated by typing "Mr Bloggs has booked in". You can also tell the system that a customer has left by typing "Mr Bloggs has left" this deletes the customer and marks all rooms booked by them as free.

In writing this programme I have tried to make it as flexible as possible. At one stage for example the system required that if multiple rooms were asked for multiple showers must be asked for as well for example: "Do you have two rooms with showers" or "Which rooms with showers are available to night" but would not allow "Which rooms with a shower are available to night". I found that when I was testing the system this became a pain as I would often use the last example. Because of this I changed the system so that it would allow a single shower with multiple rooms.

There are also cases where the system will allow slightly strange sentences. An example of this is "Do you have a single room yes for tonight" this is because the roomtype2 predicate includes [yes] and [no] for when a customer is asked if they require a shower or not. There are fairly simple ways around this but it does not adversely affect the programme so I did not correct it.

Some examples of sentences that this system will not allow all are "Do you have a room free with a shower", this system will allow "Do you have a room with shower free". The first example is tricky as the word ‘free’ is in the middle of the room definition so to allow the sentence we would have to alter the room predicate. Doing this would mean that the programme also allow sentences such as "Please book me a room free with a shower". This may not matter as a customer is unlikely to ask this but the change was not made. One other problem I encounter at was that I could not type a £ in the prologue programme so I had to use $ instead.

I could go on extending this programme forever until it accepted every single possible sentence. However this will be a mammoth task. I think that as the programme stands it will accept most questions asked by potential customers. I am pleased with how it turned out and enjoyed creating it. I believe this was a valuable exercise in Prolog programming and it taught me a lot.
Program Listing

/*  ----  Hotel  ----  */

% Call 'hotel' to run program

% You also need to load the lexical analyser, in file 'lizlex', to run the program

hotel:- 


printprompt,


readsentence(S),


reply(S).

/*  Computer's response  */

%exits program

reply(['bye']):- write('BYE'), nl.

%finds valid question, prints response.

reply(S):- 


sentence(S,[]),


hotel.

%if no valid question asks to re-phrase question.

reply(S):-


write('Please re-phrase your question'),nl,nl,


hotel.

printprompt:- write('hotel>>').

/*  ----  Grammar  ----  */

sentence --> booking, room(RT1,RT2,NI,Nu,_),


{bookroom(RT1,RT2,NI,Nu)}.

sentence --> availibility, room(RT1,RT2,_,Nu,_), free(_), night(NI),


{availiability(RT1,RT2,NI,Nu)}.

sentence --> price1, room(RT1,RT2,NI,Nu,_),


{roomprice(RT1,RT2)}.

sentence --> price2, room(RT1,RT2,NI,Nu,_), cost,


{roomprice(RT1,RT2)}.

sentence --> quword, room1(RT1,RT2,_,plur), are, free(2), night(NI),


{listrooms(RT1,RT2,NI)}.

sentence --> confirm, reservation(NI),


{confirmbooking(NI)}.

sentence --> ['Has'], name(Na,Ti), bookedin(_), night(_),


{bookedinyet(Ti,[Na])}.

sentence --> ['Has'], title(Ti), [Na], bookedin(_), night(_),


{unknownname(Ti,Na)}.

sentence --> name(Na,Ti), [has], bookedin(1),


{bookedin(Ti,[Na])}.

sentence --> title(Ti), [Na], [has], bookedin(1),


{unknownname(Ti,Na)}.

sentence --> name(Na,Ti), [has], left,


{bookedout(Ti,[Na])}.

sentence --> title(Ti), [Na], [has], left,


{unknownname(Ti,Na)}.

are --> [are].

are --> [do,you,have].

left --> [left].

left --> [booked,out].

left --> [left,the,hotel].

booking --> polite1, [book,me].

booking --> ['Can','I',have].

booking --> ['Can','I',book].

booking --> ['Please',can,'I',have].

booking --> ['I',want,to,reserve].

availibility --> ['Do',you,have].

availibility --> ['Is',there].

price1 --> polite, ['What',is,the,price,of].

price1 --> polite, ['How',much,is].

price1 --> polite, ['How',much,would,it,cost,for].

price2 --> polite, ['How',much,does].

price2 --> polite, ['How',much,would].

confirm --> polite1, [confirm].

confirm --> polite1, [check].

confirm --> [Confirm].

confirm --> [Check].

reservation(NI) --> [my,reservation], night(NI).

reservation(NI) --> [my,booking], night(NI).

reservation(NI) --> [my,reservations], night(NI).

reservation(NI) --> [my,bookings], night(NI).

reservation(NI) --> [my], room(RT1,RT2,NI,Nu,_).

bookedin(1) --> [booked,into,the,hotel].

bookedin(1) --> [booked,in].

bookedin(1) --> [arrived].

bookedin(2) --> [booked,into,the,hotel,yet].

bookedin(2) --> [booked,in,yet].

room(RT1,RT2,NI,Nu,SP) --> number(SP,Nu), room1(RT1,RT2,NI,SP).

room1(0,0,NI,SP) --> room2(SP), night(NI).

room1(RT1,0,NI,SP) --> roomtype1(RT1), room2(SP), night(NI).

%allow multiple rooms with shower - common mistake don't want to annoy customer

room1(RT1,RT2,NI,SP) --> roomtype1(RT1), room2(SP), night(NI), roomtype2(RT2,_).

room1(RT1,RT2,NI,SP) --> roomtype1(RT1), room2(SP), roomtype2(RT2,_), night(NI).

roomtype1(single) --> [single].

roomtype1(double) --> [double].

roomtype1(0) --> [].

roomtype2(yes,sing) --> [with,shower].

roomtype2(yes,sing) --> [with,a,shower].

roomtype2(yes,plur) --> [with,showers].

roomtype2(yes,_) --> [yes].

roomtype2(no,sing) --> [without,shower].

roomtype2(no,sing) --> [without,a,shower].

roomtype2(no,plur) --> [without,showers].

roomtype2(no,_) --> [no].

roomtype2(0,_) --> [].

room2(sing) --> [room].

room2(plur) --> [rooms].

number(sing,1) --> [a].

number(sing,1) --> [one].

number(plur,2) --> [two].

number(plur,3) --> [three].

night(0) --> [].

night(NI) --> [for], night1(NI).

night(NI) --> night1(NI).

night1(1) --> [tonight].

night1(1) --> [today].

night1(1) --> [this,evening].

night1(2) --> [tomorrow].

night1(2) --> [tomorrow,evening].

night1(2) --> [tomorrow,night].

cost --> [cost].

cost --> [cost,me].

polite --> polite1, polite2.

polite --> [].

polite1 --> ['Please'].

polite1 --> ['Would',you].

polite1 --> ['Could',you].

polite1 --> ['Please',would,you].

polite1 --> ['Please',could,you].

polite1 --> [].

polite2 --> [let,me,know].

polite2 --> [tell,me].

%number indicates if can be blank or not.

free(1) --> [].

free(2) --> [free].

free(2) --> [availiable].

free(2) --> [vacant].

quword --> ['What'].

quword --> ['Which'].

title(1) --> ['Mr'].

title(2) --> ['Mrs'].

title(3) --> ['Miss'].

title(4) --> ['Ms'].

name(Na,Ti,[Title|[Na|T]],T):-


title(Ti,[Title],_),


customer(Ti,[Na],_).

/*  ----  Data Base  ----  */

%roomdb(roomnumber,roomtype,shower,price).

roomdb(1,double,no,35).

roomdb(2,double,yes,40).

roomdb(3,single,yes,25).

roomdb(4,single,no,20).

roomdb(5,single,yes,25).

:-dynamic roombooking/3.

%roombooking(roomnumber,day,name)

roombooking(1,1,roomfree).

roombooking(1,2,roomfree).

roombooking(2,1,roomfree).

roombooking(2,2,roomfree).

roombooking(3,1,roomfree).

roombooking(3,2,roomfree).

roombooking(4,1,roomfree).

roombooking(4,2,roomfree).

roombooking(5,1,roomfree).

roombooking(5,2,roomfree).

:-dynamic customer/3.

%customer(title,name,bookedin).

/*  ----  Data Base Funtions  ----  */

%prints list of rooms availiable

printlist(Before,[H|T],After):-


write(Before),


write(H),


write(After), nl,


printlist(Before,T,After).

printlist(_,[],_).

%lists rooms availiable

listrooms(RT1,RT2,NI1):-


%if roomtype 0 changes it to _


chkroomtype1(1,RT1,Re1),


chkroomtype2(1,RT2,Re2),


%if night 0 askswhich night


checknight(NI1,NI2),


%gets all results in a list


findall(RMNUMBER,roomfree(RMNUMBER,Re1,Re2,NI2), Rooms),


%gets length of the list


list_length(Rooms,L),


listrooms2(Rooms,L).

%if no rooms

listrooms2(Rooms,0):-


write('no rooms are free.'),nl,nl.

%if rooms

listrooms2(Rooms,_):-


printlist(('Room '),Rooms,(' is free.')),nl.

%books a room

bookroom(RT1,RT2,NI1,Nu):-


%if type or room not specified asks them to be entered.


chkroomtype1(2,RT1,Re1),


chkroomtype2(2,RT2,Re2),


checknight(NI1,NI2),


!,


bookroom2(Re1,Re2,NI2,Nu).

%if room is availiable asks for customer name

bookroom2(Re1,Re2,NI,Nu):-


roomfree(RMNUMBER,Re1,Re2,NI),


getname(Ti,Na),


assert(customer(Ti,Na,no)),


bookroom3(Re1,Re2,NI,Na,Nu).

%if no rooms availiable tells customer

bookroom2(Re1,Re2,NI2,Nu):-


write('sorry, we were unable to complete your request.'), nl,


write('We did not have sufficient rooms availiable'), nl,nl.

%when room list is empty

bookroom3(Re1,Re2,NI,Na2,0).

%books rooms in list one at a time

bookroom3(Re1,Re2,NI,Na2,Nu):-


Nu \== 0,


roomfree(RMNUMBER,Re1,Re2,NI),


retract(roombooking(RMNUMBER,NI,_)),


assert(roombooking(RMNUMBER,NI,Na2)),


write('You have been booked into room '),


night1(NI,[NIGHT,_],_),


write(RMNUMBER),


write(' for '),


write(NIGHT),


write('.'),nl,nl,


Nu2 is Nu - 1,


bookroom3(Re1,Re2,NI,Na2,Nu2).

%if above fails then can't book all rooms, tell customer.

bookroom3(Re1,Re2,NI,Na2,Nu):-


write('sorry, we were unable to book all of your rooms.'), nl,


write('We did not have sufficient rooms availiable'), nl,nl.

%gets room price.

roomprice(RT1,RT2):-


%gets input if supplied as 0


chkroomtype1(2,RT1,Re1),


chkroomtype2(2,RT2,Re2),


%gets price


roomdb(_,Re1,Re2,Price),


%can't get pound synbol for some reason.


%displays price


write('That room costs $'),


write(Price),


write('.'),nl,


roomprice2(Re1,Re2).

%if one availiable tonight tells customer

roomprice2(RT1,RT2):-


roomfree(_,RT1,RT2,1),


write('We have one availiable tonight.'),nl,nl.

roomprice2(RT1,RT2):-


nl.

%checks for room availiability

availiability(RT1,RT2,NI,Nu):-


%validates input


chkroomtype1(1,RT1,Re1),


chkroomtype2(1,RT2,Re2),


checknight(NI,NI2),


availiability2(Re1,Re2,NI2,Nu).

%room free

availiability2(RT1,RT2,NI,Nu):-


roomfree(_,RT1,RT2,NI),


write('Yes, we have a room availiable.'),nl,nl.

%room not free

availiability2(RT1,RT2,NI,Nu):-


write('No, Im sorry but all our rooms of that type are full.'),nl,nl.

%confirms customer booking

confirmbooking(NI):-


checknight(NI,NI2),


%gets name


getname(Ti,Na),


%gets all bookings


findall(RoomNumber,roombooking(RoomNumber,NI2,Na),Bookings),


list_length(Bookings,NumB),


confirmbooking2(Bookings,NumB,NI2).

%no bookings

confirmbooking2(Bookings,0,NI):-


write('You have no reservations.'),nl,nl.

%lists bookings

confirmbooking2(Bookings,NumB,NI):-


NumB \== 0,


night1(NI,[Night,_],_),


write('Your reservations for '),


write(Night),


write(' are listed below:'),nl,


printreslist(Bookings),nl.

%if unknown name entered

unknownname(Ti,Na):-


write('I do not know '),


title(Ti,[Title,_],_),


write(Title),


write(' '),


write(Na),


write('.'),nl,nl.

%has a customer booked in yet

bookedinyet(Ti,Na):-


%consult db


customer(Ti,Na,yes),


%writes yes


write('Yes '),


title(Ti,[Title,_],_),


write(Title),


write(' '),


namefromlist(Na,Name),


write(Name),


write(' has booked in.'),nl,nl,!.

%not booked in

bookedinyet(Ti,Na):-


customer(Ti,Na,no),


write('No '),


title(Ti,[Title,_],_),


write(Title),


write(' '),


namefromlist(Na,Name),


write(Name),


write(' has not booked in yet.'),nl,nl,!.

%marks customer as booked in

bookedin(Ti,Na):-


retract(customer(Ti,Na,_)),


assert(customer(Ti,Na,yes)),


title(Ti,[Title,_],_),


write(Title),


write(' '),


namefromlist(Na,Name),


write(Name),


write(' booked in.'),nl,nl.

%deletes name and all reservations

bookedout(Ti,Na):-


retract(customer(Ti,Na,_)),


findall([RmNumber,Ni],roombooking(RmNumber,Ni,Na),Bookings),


bookedout2(Bookings),


title(Ti,[Title,_],_),


write(Title),


write(' '),


namefromlist(Na,Name),


write(Name),


write(' has been deleted. Rooms marked as free.'),nl,nl.

bookedout2([]).

bookedout2([[RmNumber|[Ni]]|T]):-


retract(roombooking(RmNumber,Ni,_)),


assert(roombooking(RmNumber,Ni,roomfree)),


bookedout2(T).

namefromlist([Na],Na).


printreslist([]).

printreslist([H|T]):-


roomdb(H,Type1,Type2,Price),


write('Room '),


write(H),


write(', '),


write(Type1),


write(', with shower: '),


write(Type2),


write(', $'),


write(Price),


write('.'),nl,


printreslist(T).

roomfree(RMNUMBER,Type1,Type2,Day):-


roomdb(RMNUMBER,Type1,Type2,_),


roombooking(RMNUMBER,Day,roomfree).

%gets users name

getname(Ti,Na2):-


write('please enter your name(e.g. Mr Bloggs.)'), nl,


readsentence(Na1),


title(Ti,Na1,Na2),!.

%repeats until valid name entered

getname(Ti,Na):-


getname(Ti,Na).

%roomtype_(getif0,typenumber,description)

chkroomtype1(_,single,single).

chkroomtype1(_,double,double).

chkroomtype1(1,0,_).

chkroomtype1(2,0,RmType):-


write('please enter roomtype (single/double)'), nl,


readsentence(RMTYPE),


%check input against grammar (single/double entered)


roomtype1(RmType,RMTYPE,_).

%repeat if not found the first time(invalid input)

chkroomtype1(2,0,RMTYPE):-


chkroomtype1(2,0,RMTYPE).

chkroomtype2(_,yes,yes).

chkroomtype2(_,no,no).

chkroomtype2(1,0,_).

chkroomtype2(2,0,RmType):-


write('room with a shower (yes,no)'), nl,


readsentence(RMTYPE),


%check grammar


roomtype2(Number,_,RMTYPE,_).

chkroomtype2(2,0,RMTYPE):-


chkroomtype2(2,0,RMTYPE).

nightinput(In,Ni,roomfree):-


In == Ni,!.

nightinput(In,Ni,_):-


In \== Ni,!.

checknight(0,NIo):-


write('please enter night(tonight/tomorrow)'), nl,


readsentence(In),


night1(NIo,In,_).

checknight(0,NIo):-


checknight(0,NIo).

checknight(NI,NI):-


NI \==  0.

list_length([], 0).

list_length([_|T], N):-


list_length(T, N1),


N is N1 + 1.

User-Machine Dialogue

| ?- hotel.

hotel>>|: Please book me a single room without shower.

please enter night(tonight/tomorrow)

|: tonight.

please enter your name(e.g. Mr Bloggs.)

|: Mr Bloggs.

You have been booked into room 4 for tonight.

hotel>>|: I want to reserve a room for tonight.

please enter roomtype (single/double)

|: double.

room with a shower (yes,no)

|: yes.

please enter your name(e.g. Mr Bloggs.)

|: Mr Maguire.

You have been booked into room 1 for tonight.

hotel>>|: Please book me three rooms for tonight.

please enter roomtype (single/double)

|: single.

room with a shower (yes,no)

|: with showr.

room with a shower (yes,no)

|: with shower.

please enter your name(e.g. Mr Bloggs.)

|: Mr Roadnight.

You have been booked into room 3 for tonight.

You have been booked into room 5 for tonight.

sorry, we were unable to book all of your rooms.

We did not have sufficient rooms availiable

hotel>>|: Do you have a double room free for this evening.

Yes, we have a room availiable.

hotel>>|: What is the price of a single room with a shower.

That room costs $25.

We have one availiable tonight.

hotel>>|: How much is a room.

please enter roomtype (single/double)

|: double.

room with a shower (yes,no)

|: yes.

That room costs $35.

We have one availiable tonight.

hotel>>|: Is there a single room availiable for tomorrow night.

Yes, we have a room availiable.

hotel>>|: Do you have a room free tonight.

Yes, we have a room availiable.

hotel>>|: Do you have a single room free tonight.

No, Im sorry but all our rooms of that type are full.

hotel>>|: Please confirm my reservation for tonight.

please enter your name(e.g. Mr Bloggs.)

|: Mr Roadnight.

Your reservations for tonight are listed below:

Room 3, single, with shower: yes, $25.

Room 5, single, with shower: yes, $25.

hotel>>|: Please would you confirm my booking for tonight.

please enter your name(e.g. Mr Bloggs.)

|: Mr Larson.

You have no reservations.

hotel>>|: Which rooms are vacant tonight.

Room 2 is free.

hotel>>|: Which rooms are free tomorrow night.

Room 1 is free.

Room 2 is free.

Room 3 is free.

Room 4 is free.

Room 5 is free.

hotel>>|: Which single rooms are free tomorrow.

Room 3 is free.

Room 4 is free.

Room 5 is free.

hotel>>|: Which single rooms with a shower are free tomorrow.

Room 3 is free.

Room 5 is free.

hotel>>|: Has Mr Roadnight booked into the hotel for tonight.

No Mr Roadnight has not booked in yet.

hotel>>|: Mr Roadnight has booked in.

Mr Roadnight booked in.

hotel>>|: Has Mr Roadnight booked in yet.

Yes Mr Roadnight has booked in.

hotel>>|: Mr Roadnight has left.

Mr Roadnight has been deleted. Rooms marked as free.

hotel>>|: Has Mr Roadnight booked in.

I do not know Mr Roadnight.

hotel>>|: What rooms are free tonight.

Room 2 is free.

Room 3 is free.

Room 5 is free.

